
Stan code for 2-parameter hierarchical IRT model  

 

data { 
  int<lower=1> I;               // # questions 
  int<lower=1> J;               // # persons 
  int<lower=1> N;               // # observations 
  int<lower=1, upper=I> ii[N];  // question for n 
  int<lower=1, upper=J> jj[N];  // person for n 
  int<lower=0, upper=1> y[N];   // response for n 
  int<lower=1, upper=2> psychosis[J]; // status 
} 
 
parameters { 
  vector[2] xi[I];              // alpha/beta pair vectors 
  vector[2] mu;                 // vector for alpha/beta means 
  vector<lower=0>[2] tau;       // vector for alpha/beta residual sds 
  cholesky_factor_corr[2] L_Omega; 
  vector[2] psychosis_int;        // effect of psychosis on severity 
  vector[J] epsilon;              // severity for person j 
} 
 
transformed parameters { 
  vector[J] severity; 
  vector[I] discriminativeness; 
  vector[I] difficulty; 
  for (j in 1:J){  
    severity[j] =  psychosis_int[psychosis[j]] + epsilon[j];  
  } 
  for (i in 1:I) { // rename variables for readability 
    discriminativeness[i] = xi[i,1]; 
    difficulty[i] = xi[i,2]; 
  } 
} 
 
model { 
  matrix[2,2] L_Sigma; 
  L_Sigma = diag_pre_multiply(tau, L_Omega); 
  for (i in 1:I){ 
    xi[i] ~ multi_normal_cholesky(mu, L_Sigma); 
  } 
  L_Omega ~ lkj_corr_cholesky(4); 
  mu[1] ~ normal(0,1); 
  tau[1] ~ normal(0,1); 
  mu[2] ~ normal(0,5); 
  tau[2] ~ normal(0,1); 
   
  psychosis_int ~ normal(0,.5); 
  epsilon ~ normal(0,.5); 
   
  y ~ bernoulli_logit( (discriminativeness[ii] .* severity[jj])  - difficulty[ii]); 
} 
 
generated quantities { 
  corr_matrix[2] Omega; 
  Omega = multiply_lower_tri_self_transpose(L_Omega); 
} 


